
• More on addressing modes. 
• 9S12 cycles and execution time. 
• AS12 Assembler Directives 
• Huang, Sections 1.6 through 1.10 

o Using X and Y registers as pointers 
o How to tell which branch instruction to use 
o How to hand assemble a program 
o Number of cycles and time taken to execute an 9S12 program 

The HCS12 has 6 addressing modes

Most of the HC12’s instructions access data in memory
There are several ways for the HC12 to determine which address to access

Effective Address:
Memory address used by instruction

ADDRESSING MODE:
How the HC12 calculates the effective address

HC12 ADDRESSING MODES:
INH Inherent
IMM Immediate
DIR Direct
EXT Extended
REL Relative (used only with branch instructions)
IDX Indexed (won’t study indirect indexed mode)

Using X and Y as Pointers
• Registers X and Y are often used to point to data.
• To initialize pointer use

ldx #table
not

ldx table

• For example, the following loads the address of table ($2000) into X; i.e., X will point
to table:

ldx #table ; Address of table ⇒ X

The following puts the first two bytes of table ($0C7A) into X. X will not point to table:
ldx table   ; First two bytes of table ⇒ X

• To step through table, need to increment pointer after use

ldaa 0,x



inx
or

ldaa 1,x+

table

0C
7A
D5
00
61
62
63
64

org $2000
table: dc.b 12,122,-43,0

dc.b ‘a’,’b’,’c’,’d’

Which branch instruction should you use?
Branch if A > B
Is 0xFF > 0x00?

If unsigned, 0xFF = 255 and 0x00 = 0,
so 0xFF > 0x00

If signed, 0xFF = −1 and 0x00 = 0,
so 0xFF < 0x00

Using unsigned numbers: BHI (checks C bit of CCR) 
Branch if Higher (if C + Z = 0 )(unsigned)

Using signed numbers: BGT (checks V bit of CCR)
Branch if Greater Than (if Z + (N ⊕ V) = 0) (signed)

For unsigned numbers, use branch instructions which check C bit
For signed numbers, use branch instructions which check V bit

Hand Assembling a Program
To hand-assemble a program, do the following:

1. Start with the org statement, which shows where the first byte of the program will go
into memory.
(e.g., org $2000 will put the first instruction at address $2000.)



2. Look at the first instruction. Determine the addressing mode used.
(e.g., ldab #10 uses IMM mode.)

3. Look up the instruction in the HCS12 Core Users Guide, find the appropriate 
Addressing Mode, and the Object Code for that addressing mode.
(e.g., ldab IMM has object code C6 ii.)

Table 5.1 of the Core Users Guide has a concise summary of the instructions, 
addressing modes, op-codes, and cycles.

4. Put in the object code for the instruction, and put in the appropriate operand. Be
careful to convert decimal operands to hex operands if necessary.
(e.g., ldab #10 becomes C6 0A.)

5. Add the number of bytes of this instruction to the address of the instruction to 
determine
the address of the next instruction.
(e.g., $2000 + 2 = $2002 will be the starting address of the next instruction.)

          org $2000
          ldab #10
loop: clra
          dbne b,loop
          swi











68HC12 Cycles

• 68HC12 works on 48 MHz clock
• A processor cycle takes 2 clock cycles – P clock is 24 MHz
• Each processor cycle takes 41.7 ns (1/24 μs) to execute
• An instruction takes from 1 to 12 processor cycles to execute
• You can determine how many cycles an instruction takes by looking up the CPU cycles
for that instruction in the Core Users Guide.
– For example, LDAA using the IMM addressing mode shows one CPU cycle (of type
P).
– LDAA using the EXT addressing mode shows three CPU cycles (of type rPf).
– Section A.27 of the Core Users Guide explains what the HCS12 is doing during
each of the different types of CPU cycles.

2000 org $2000 ; Inst Mode Cycles
2000 c6 0a ldab #10 ; LDAB (IMM) 1
2002 87 loop: clra ; CLRA (INH) 1
2003 04 31 fc dbne b,loop ; DBNE (REL) 3
2006 3f swi ; SWI 9

The program executes the ldab #10 instruction once (which takes one cycle). It then goes
through loop 10 times (which has two instructions, on with one cycle and one with three
cycles), and finishes with the swi instruction (which takes 9 cycles).

Total number of cycles:

1 + 10 × (1 + 3) + 9 = 50

50 cycles = 50 × 41.7 ns/cycle = 2.08 μs





Assembler Directives

• In order to write an assembly language program it is necessary to use assembler 
directives.
• These are not instructions which the HC12 executes but are directives to the assembler 
program about such things as where to put code and data into memory.
• We will use only a few of these directives. (Note: In the following table, [] means an 
optional argument.) Here are the ones we will need:

Directive Name Description Example

equ Give a value to a symbol len: equ 100
org Set starting value of 

location counter where code 
or data will go

org $1000

dc[.size] Allocate and initialize 
storage  for variables. Size 
can be b (byte) or w (two 
bytes)
If no size is specified, b is 
used

var: dc.b 2,18

ds[.size] Allocate specified number 
of  storage spaces. size is 
the same as for dc directive

table: ds.w 10

fcc Encodes a string  of ASCII 
characters. The first 
character is the delimiter. 
The string terminates at the 
next occurrence of the 
delimiter

table: fcc "Hello"

Using labels in assembly programs
A label is defined by a name followed by a colon as the first thing on a line.
When the label is referred to in the program, it has the numerical value of the location 
counter when the label was defined.

Here is a code fragment using labels and the assembler directives dc and ds:

org $2000
table1:dc.b $23,$17,$f2,$a3,$56
table2: ds.b 5
var: dc.w $43af



The as12 assembler produces a listing file (.lst) and a symbol file (.sym). Here is the 
listing file from the assembler:

as12, an absolute assembler for Motorola MCU's, version 1.2e

2000                    org $2000
2000 23 17 f2 a3 56     table1: dc.b $23,$17,$f2,$a3,$56
2005                    table2: ds.b 5
200a 43 af              var: dc.w $43af
                        
Executed: Sat Jan 15 13:19:23 2008
Total cycles: 0, Total bytes: 7
Total errors: 0, Total warnings: 0

Note that table1 is a name with the value of $2000, the value of the location counter 
defined in the org directive. Five bytes of data are defined by the dc.b directive, so the 
location counter is increased from $2000 to $2005.
table2 is a name with the value of $2005. Five bytes of data are set aside for table2 by the 
ds.b 5 directive. The as12 assembler initialized these five bytes of data to all zeros. var is 
a name with the value of $200a, the first location after table2.


